

Upgrading from VLOOKUP to XLOOKUP

XLOOKUP has been available for over five years and is designed to replace VLOOKUP entirely. This guide covers the seven key scenarios you need to know to upgrade your formulas.

CONTENTS

- 1 Exact Match Lookups
- 2 Error Handling
- 3 Looking to the Left
- 4 Horizontal Lookups
- 5 Two-Way Lookups
- 6 Find Last with search_mode
- 7 Closest Match
- 8 Modernize Formulas with Copilot

1 Exact Match Lookups

The Most Common Upgrade: Side-by-Side Comparison

VLOOKUP requires four arguments. XLOOKUP only needs three. The biggest conceptual shift is moving from a column index number to directly selecting the return array.

VLOOKUP

```
=VLOOKUP(A2, B2:D25, 2, FALSE)
```

XLOOKUP

```
=XLOOKUP(A2, B2:B25, C2:C25)
```

lookup_value	The value you are searching for. Same as VLOOKUP.
lookup_array	The single column (or row) to search in. Not the entire table.
return_array	The column containing the values you want returned.
if_not_found	Optional. Text to display when no match is found.
match_mode	Optional. Defaults to exact match (0). No need to specify FALSE.
search_mode	Optional. Controls search direction and binary search behavior.

Two major differences from VLOOKUP: First, you select only the lookup column, not the entire table range. Second, instead of counting a column index number, you select the return column directly. This means inserting or deleting columns no longer breaks your formulas.

KEY ADVANTAGE

Because you select the return column directly, adding or removing columns from your data will never break an XLOOKUP formula. With VLOOKUP, any column shift would require updating every formula manually.

MUSCLE MEMORY TIP

The hardest habit to break is counting columns. Train yourself to stop after selecting the lookup array and just select the return column instead of entering a number.

The most common error you will encounter is a range length mismatch, where the lookup array and return array have different numbers of rows. Using Excel Tables prevents this automatically because table columns always stay the same length.

2 Error Handling

Replace IFERROR with the Built-in if_not_found Argument

With VLOOKUP, the standard approach was to wrap the formula in IFERROR to display a friendly message when no match was found. XLOOKUP has this built in as its fourth argument, if_not_found.

VLOOKUP WITH IFERROR

```
=IFERROR(VLOOKUP(A2, B2:D25, 2, FALSE), "Not found")
```

XLOOKUP BUILT-IN

```
=XLOOKUP(A2, B2:B25, C2:C25, "Not found")
```

An important distinction: if_not_found only handles cases where there is no matching value. It does not suppress range length errors or other formula errors. This is intentional behavior.

WHY THIS IS A GOOD THING

Errors like mismatched range lengths indicate real problems in your data or formula that need to be fixed, not hidden. IFERROR would silently mask these issues.

PRO TIP

Write the formula without the if_not_found argument first. Let it return errors so you can see exactly what is happening. Once the formula works correctly, add the argument to handle the no-match case.

PRO TIP: UNEXPECTED ZEROS

If your formula returns a zero instead of a blank, this means the matching cell in the return array is empty. The formula is working correctly. The fix is to enter a value in that blank source cell rather than adjusting the formula.

3 Looking to the Left

Replace *INDEX/MATCH* for Reverse Lookups

VLOOKUP can only look to the right. If your lookup column was to the right of your return column, you had to use the more complex INDEX/MATCH combination. XLOOKUP has no directional restriction.

INDEX/MATCH (LEFT LOOKUP)

```
=INDEX (A2:A25, MATCH (G2, C2:C25, 0))
```

XLOOKUP (ANY DIRECTION)

```
=XLOOKUP (G2, C2:C25, A2:A25)
```

The return array simply points to whichever column you need, regardless of whether it is to the left or right of the lookup column.

4 Horizontal Lookups

Replace *HLOOKUP* for Row-Based Searches

VLOOKUP searches vertically through columns; HLOOKUP searches horizontally through rows. XLOOKUP handles both. Instead of selecting a column range, you select a row range for your lookup and return arrays.

HLOOKUP

```
=HLOOKUP (A2, B2:Z3, 2, FALSE)
```

XLOOKUP

```
=XLOOKUP (A2, B2:Z2, B3:Z3)
```

The syntax is identical to a vertical lookup. The only difference is that your lookup and return arrays are rows instead of columns. XLOOKUP detects the orientation automatically.

5 Two-Way Lookups

Replace *VLOOKUP/MATCH* or *INDEX/MATCH* Combinations

A two-way lookup finds a value by matching both a row and a column. The XLOOKUP approach is to nest one XLOOKUP inside another: the inner formula locates the correct column, and the outer formula locates the correct row.

Step 1: Use XLOOKUP to match on the column headers. This returns an entire column as a spilled range.

```
=XLOOKUP(G2, B1:E1, B2:E25)
```

Step 2: Wrap that formula in another XLOOKUP to match the row value, returning a single cell.

```
=XLOOKUP(G3, A2:A25, XLOOKUP(G2, B1:E1, B2:E25))
```

PRO TIP: RETURN MULTIPLE COLUMNS

XLOOKUP can return an entire spilled range by selecting multiple adjacent columns as the return array. This is useful when you want to pull several related fields in a single formula.

6 Find Last with search_mode

A Task VLOOKUP Simply Could Not Do

VLOOKUP always finds the first match and has no way to find the last match in a list without sorting the data. XLOOKUP includes a `search_mode` argument that controls direction.

1	Search first to last (default)
-1	Search last to first — finds the last match
2	Binary search, ascending order
-2	Binary search, descending order

```
=XLOOKUP(A2, B2:B25, C2:C25, , , -1)
```

NEED MULTIPLE MATCHES?

XLOOKUP always returns a single result. If you need to return all matching values, use the FILTER function instead.

7 Closest Match

Approximate Match Lookups for Tiered Data

VLOOKUP's fourth argument accepts `TRUE` for an approximate (closest) match, which finds the largest value less than or equal to the lookup value. XLOOKUP uses the `match_mode` argument to achieve the same result, giving you more explicit control.

VLOOKUP (APPROXIMATE MATCH)

```
=VLOOKUP(C4, H7:J11, 3, TRUE)
```

XLOOKUP (MATCH_MODE = -1)

```
=XLOOKUP(C4, H7:H11, J7:J11, , -1)
```

In the example above, a sales amount of \$37,500 is looked up in a commission rate table. VLOOKUP with `TRUE` requires the data to be sorted ascending to work correctly. XLOOKUP with `match_mode = -1` finds the exact match or next smaller value, and does not require sorting.

0	Exact match (default)
-1	Exact match or next smaller value — use for tiered rate tables
1	Exact match or next larger value
2	Wildcard match using *, ?, and ~

VLOOKUP SORTING REQUIREMENT

VLOOKUP approximate match (`TRUE`) requires your lookup column to be sorted in ascending order or it will return incorrect results. XLOOKUP `match_mode` does not have this requirement.

8 Modernize Formulas with Copilot

Let AI Handle the Conversion for You

If you have a workbook full of existing VLOOKUP formulas, you can ask Copilot in Excel to convert them to XLOOKUP automatically. Select a cell containing a VLOOKUP formula and prompt Copilot to rewrite it.

EXAMPLE COPILOT PROMPT

Replace the VLOOKUP formula in this cell with XLOOKUP. Use the `if_not_found` argument instead of wrapping it in IFERROR.

Copilot can handle the argument translation and is especially useful for the trickier conversions like two-way lookups or formulas with IFERROR wrappers.

I encourage you to keep using and practicing XLOOKUP and it will become easier in no time.

Watch the YouTube tutorial and read the blog post at:

<https://www.excelcampus.com/functions/xlookup-vs-vlookup-complete-guide/>